

Прямые методы математической физики с использованием системы Maple

Д.т.н., профессор Д.П. Голоскоков
Государственный университет морского и речного флота
имени адмирала С.О. Макарова
Кафедра прикладной математики
GoloskokovDP@gumrf.ru

- * Под прямыми методами приближенного решения задач математической физики понимают такие
- * методы, которые сводят эти задачи к конечным системам алгебраических уравнений. Такое определение дает С. Г. Михлин в своей книге "Прямые методы в математической физике" (ГИТТЛ, М.-Л., 1950).
- * Прямые методы широко используются в практике инженерных расчетов, значительна роль этих методов и в теоретических исследованиях.
- * Во многих случаях краевую задачу для дифференциального уравнения можно заменить равносильной задачей об отыскании функции, доставляющей некоторому интегралу наименьшее значение. Задачи такого типа называются вариационными, а методы, позволяющие свести задачу об интегрировании дифференциального уравнения к равносильной вариационной задаче, называются вариационными методами.

- * Математически проблема решения краевой задачи для дифференциального уравнения эквивалентна задаче вариационного исчисления — о минимуме интеграла, для которого данное дифференциальное уравнение служит уравнением Эйлера.
- * Для решения задач вариационного исчисления наряду с уравнением Эйлера могут применяться непосредственные, прямые методы, которые обходят применение дифференциальных уравнений; благодаря указанной эквивалентности, эти методы являются одновременно и методами решения краевых задач для дифференциальных уравнений. В основе этих методов лежит идея построить искомую функцию, дающую экстремум функционала, при помощи некоторого предельного процесса, исходя непосредственно из вида того интеграла, экстремум которого ищется.
- * Наиболее известными среди таких методов являются методы Рэля-Ритца и Бубнова-Галеркина.
- * В настоящей лекции мы покажем использование системы символьных вычислений Maple для реализации методов приближенного решения дифференциальных уравнений, связанных с вариационным исчислением.

1. Простейшая вариационная задача. Конечно-разностный метод Эйлера

Мы начнем рассмотрение с простейшей вариационной задачи для одномерного интеграла: найти функцию $y(x)$, доставляющую экстремум (или просто стационарное значение) функционалу

$$J[y(x)] = \int_a^b F(x, y, y') dx,$$

где F — заданная функция трех аргументов, непрерывная вместе с производными второго порядка в некоторой области. Значения искомой функции $y(x)$ на концах промежутка интегрирования заданы:

$$y(a) = y_a, y(b) = y_b.$$

Будем решать задачу одним из прямых методов — конечно-разностным методом Эйлера. Значения функционала $J[y(x)]$ станем рассматривать не на произвольных, допустимых в данной вариационной задаче кривых, а лишь на ломаных, составленных из заданного числа N прямолинейных звеньев, с заданными абсциссами вершин

$$a, a+h, a+2h, \dots, a+(N-1)h, a+Nh, h = \frac{b-a}{N}.$$

На этих ломаных функционал превращается в функцию $\Phi(Y_1, Y_2, \dots, Y_{N-1})$ ординат Y_1, Y_2, \dots, Y_{N-1} вершин ломаной.

Ординаты Y_1, Y_2, \dots, Y_{N-1} выбираются так, чтобы функция $\Phi(Y_1, Y_2, \dots, Y_{N-1})$ достигала экстремума, т. е. они определяются из системы уравнений

$$\frac{\partial \Phi}{\partial Y_k} = 0, k = 1, 2, \dots, N - 1,$$

ординаты Y_0 и Y_N известны из граничных условий $y(a) = y_a, y(b) = y_b$. Это типичный метод дискретизации, который состоит в приближенной замене функционала $J[y(x)]$ на сумму

$$I_N = \sum_{k=1}^{N-1} F\left(X_k, Y_k, \frac{Y_{k+1} - Y_k}{h}\right),$$

зависящую от конечного числа дискретных значений $Y_k = y(X_k)$ искомой функции в заданных узлах $X_k = a + kh$.

Пример 1. Найти экстремаль функционала

$$\int_0^1 \left[\left(\frac{dy}{dx} \right)^2 + y^2 + 2xy \right] dx, y(0) = 0, y(1) = 0.$$

Решение. Реализацию намеченного алгоритма решения поставленной задачи выполним в системе аналитических вычислений Maple:

> restart; interface(displayprecision=3):

Аппроксимация подынтегральной функции конечными разностями:

> F := (Y, m, h) -> (Y[m+1]-Y[m])^2/h^2+Y[m]^2+2*X[m]*Y[m] :

Интеграл заменяем суммой по формуле прямоугольников:

```
> S1 := (h, F, N) -> h*(sum(F(Y, i, h), i = 0 .. N-1)):
```

Задаем пределы интегрирования

```
> a := 0: b := 1:
```

Выбираем число узловых точек и определяем шаг интегрирования

```
> N := 10: h := (b-a)/N:
```

Вычисляем абсциссы вершин ломаной

```
> for j from 0 to N do X[j] := h*j+a end do:
```

Функционал как функция ординат вершин ломаной

```
Phi := S1(h, F, N):
```

Учет граничных условий

```
> Y[0] := 0; Y[N] := 0:
```

Составление минимизирующей системы уравнений

```
> for k to N-1 do eq[k] := evalf(diff(Phi, Y[k])) = 0 end do:
```

```
var := {}; eqns := {};
```

```
for k to N-1 do var := var union {Y[k]};
```

```
eqns := eqns union {eq[k]} end do:
```

Учет граничных условий

```
> Y[0] := 0; Y[N] := 0:
```

Решение системы

```
> res := solve(eqns, var): assign(res):
```


Формирование списка точек вершин ломаной

```
> for j from 0 to N do P[j] := [X[j], Y[j]] end do:
```

```
L := [seq(P[k-1], k = 1 .. N+1)]:
```

Построение графика решения

```
> plot(L, x = 0 .. 1, title = cat("Число узлов N = ", convert(N, string)),  
titlefont = ["ROMAN", 15], legend = 'метод Эйлера',  
font = [Courier, roman, 14], labelfont = [Helvetica, roman, 14], gridlines =  
true);
```



Для сравнения получим точное решение задачи

```
> with(VariationalCalculus):
```

```
f := (diff(y(x), x))^2+y(x)^2+2*x*y(x):
```

```
ode := EulerLagrange(f, x, y(x)):
```

```
problem :=
```

```
ode union {y(0) = 0, y(1) = 0}: dsolve(problem, y(x));
```

$$y(x) = \frac{e^{-x}}{e^{-1} - e} - \frac{e^x}{e^{-1} - e} - x$$

```
y := unapply(rhs(%), x):
```

Построим графики приближенного и точного решений

```
> plot([L, y(x)], x = 0 .. 1,
```

```
title = cat("Число узлов N = ",
```

```
convert(N, string)),
```


```
titlefont = ["ROMAN", 15],
```

```
legend = ["метод Эйлера",
```

```
"Точное решение"],
```

```
style = [point, line],
```

```
labelfont = [Helvetica, roman, 14], gridlines = true);
```



2. Простейшая вариационная задача. Метод Ритца

Метод Ритца представляет собой один из методов построения минимизирующей последовательности.

Пусть функция $u(P)$ удовлетворяет в некоторой области Ω неоднородному уравнению $Au = f(P)$, и некоторым однородным граничным условиям. Мы знаем, что решение этого уравнения, где A — положительный оператор, сводится к нахождению минимума функционала

$$F[u(P)] = (Au, u) - 2(u, f) = \int_{\Omega} [u(P)Au(P) - 2u(P)f(P)]d\Omega \quad (1)$$

Эту последнюю задачу будем приближенно решать следующим образом. Выберем последовательность функций

$$\varphi_1(P), \varphi_2(P), \dots, \varphi_n(P), \dots, \quad (2)$$

принадлежащих области определения оператора A . Подчиним эти функции двум условиям: 1) последовательность полна по энергии; 2) для любого n функции $\varphi_1(P), \varphi_2(P), \dots, \varphi_n(P)$ линейно независимы. Эти функции называют координатными, или базисными функциями.

Возьмем линейную комбинацию первых n координатных функций:

$$u_n(P) = \sum_{k=1}^n a_k \varphi_k(P) \quad (3)$$

где a_k — произвольные числа. Подставим (3) вместо $u(P)$ в функционал (1); это превратит функционал $F[u(P)]$ в функцию n независимых переменных $a_k, k = 1, 2, \dots, n$:

$$F[u_n] = \sum_{k=1}^n \sum_{j=1}^n (A\varphi_k, \varphi_j) a_k a_j - 2 \sum_{k=1}^n (\varphi_k, f) a_k. \quad (4)$$

Выберем коэффициенты a_k так, чтобы функция (4) приняла минимальное значение. Необходимые условия минимума функции (4), как известно, такие:

$$\frac{\partial F[u_n]}{\partial a_i} = 0, i = 1, 2, \dots, n \quad (5)$$

Отметим, что используя положительность оператора A , можно доказать, что коэффициенты a_k , удовлетворяющие системе (5), реализуют минимум величины $F[u_n]$. Выпишем явно систему (5)

$$\sum_{k=1}^n (A\varphi_i, \varphi_k) a_k = (\varphi_i, f), i = 1, 2, \dots, n. \quad (6)$$

Определитель системы (6) есть так называемый определитель Грамма линейно независимых функций, поэтому он отличен от нуля. Отсюда следует, что система (6) всегда разрешима, если оператор A — положительный. Найдя коэффициенты a_k из системы (6) и подставив их в (3), получим функцию $u_n(P)$, которую называют приближенным решением уравнения $Au = f(P)$ по Ритцу.

*** Справедлива теорема:** приближенные решения по Ритцу образуют минимизирующую последовательность для функционала (1), если только это уравнение имеет решение с конечной энергией.

Следовательно, приближенные решения по Ритцу сходятся по энергии к точному решению. Они сходятся к точному решению и в среднем, если A — положительно определенный оператор.

Пример 2. Для сравнения решим методом Ритца задачу из примера 1.

> restart;

Точное решение

> y:=x->exp(-x)/(exp(-1)-exp(1))-exp(x)/(exp(-1)-exp(1))-x;

Выбираем базисные функции и определяем аппроксимирующую функцию

> phio:=x->y1+(y2-y1)*(x-x1)/(x2-x1):

```
> Up:=proc(x,N)option operator,arrow; local n;
  phio(x)+(x-x1)*(x-x2)*sum(a[n]*x^n,'n'=0..N);
end proc:
```

Вводим граничные точки и задаем количество членов ряда

```
> x1:=0: x2:=1: y1:=0: y2:=0
```

Задаем число аппроксимирующих функций и решаем задачу

```
> N:=3: c0:=`cross`: c1:=`circle`: c2:=`box`: c3:=`diamond`:
```

```
  y:='y': F:=diff(y(x),x)^2+y(x)^2+2*x*y(x):
```

```
  for j from 0 to N do
```

```
    a:=array(0..j): u:=Up(x,j): end do:
```

```
  Fu:=simplify(int(subs(y(x)=u,F),x=x1..x2)):
```


```
> with(Optimization): Minimize(Fu);
```

```
  [-0.02030, [a0 = 0.14907, a1 = 0.14926, a2 = 0.00660, a3 = 0.00809]]
```

```
> assign(%[2]):
```

Для сравнения изобразим на одном рисунке точное и приближенное решения

```
> pUp_3:=plot(Up(x,3),x=x1..x2,color=blue,style=point,numpoints=1,symbol=c3,
  legend=`Метод Рунца N = 3`): plots[display]({pUp_3,pr1});
```



Формула приближенного решения

$U_p(x, 3);$

$$x(x - 1)(0.14907 + 0.14925x + 0.00660x^2 + 0.00809x^3)$$

Как видим, здесь результат лучше, чем в примере 1 — лучшее соответствие точному решению при меньшем числе N .

3. Задача Дирихле в прямоугольнике. Метод Бубнова-Галеркина

Пусть функция $u(P)$ удовлетворяет в некоторой области Ω неоднородному уравнению $Au = f(P)$ и некоторым однородным граничным условиям. Здесь мы не будем предполагать положительность оператора A . Выберем бесконечную последовательность координатных функций (2), которые достаточное число раз (в соответствии с данными задачи) непрерывно дифференцируемы в замкнутой области $\Omega = \Omega + S$, где S — граница области Ω . Будем считать, что эти функции удовлетворяют всем краевым условиям задачи. Будем считать также, что как уравнение $Au = f(P)$, так и краевые условия — линейные. Составим функцию (3), где a_k — произвольные числа. Эта функция удовлетворяет всем краевым условиям задачи. Подставим теперь (3) в уравнение $Au = f(P)$; получим невязку в выполнении уравнения

$$Au_n - f(P) = 0. \quad (7)$$

По методу Бубнова-Галеркина коэффициенты a_k определяются из требования, чтобы левая часть (7) была ортогональна функциям $\varphi_1(P), \varphi_2(P), \dots, \varphi_n(P)$. Таким образом, метод Бубнова-Галеркина приводит к системе линейных алгебраических уравнений (6), по виду тождественной системе, к которой приводит метод Ритца.

Отсюда заключаем, что методы Бубнова-Галеркина и Ритца совпадают, если ператор A — положительно определенный. В общем же случае метод Ритца неприменим, тогда как метод Бубнова-Галеркина сохраняет силу.

Рассмотрим следующую задачу Дирихле для уравнения Пуассона: требуется найти функцию $u(x, y)$, удовлетворяющую уравнению Пуассона — $\Delta u = f(x, y)$ в прямоугольнике и условию $u|_{\Gamma} = 0$ на границе этого прямоугольника — граничному условию первого рода. Эта задача встречается во многих приложениях математической физики. В механике жидкости и газа она описывает стационарное течение вязкой жидкости в канале с поперечным сечением в форме прямоугольника; в электростатике — потенциал электростатического поля; в теории теплопроводности — стационарное распределение температуры; в статике упругих тел — равновесие прямоугольной мембраны, кручение стержня с поперечным сечением в форме прямоугольника и многие другие процессы.

Пример 3. Рассмотрим модельную задачу о стационарном течении вязкой жидкости в канале с поперечным сечением в форме квадрата.

Уравнение для компоненты скорости, направленной вдоль канала, приведенное к безразмерной форме за счет надлежащего выбора масштабов, имеет вид

$$\Delta u = -1; \text{ граничные условия — } u|_{x=\pm 1} = 0, u|_{y=\pm 1} = 0.$$

Решение. Продемонстрируем применение метода Бубнова-Галеркина к решению этой задачи в системе Maple.

```
> restart;
```

```
> interface(displayprecision = 5):
```

Уравнение Пуассона определим как:

```
> pde:=diff(u(x,y),x$2)+diff(u(x,y),y$2)+1;
```

Подготовим процедуру, реализующую алгоритм метода Бубнова-Галеркина:

```
> Bubnov_Gal:=proc(pde,n,a,bol::boolean) local eqns,vars,sol,i;
  eqns:={}; vars:={}; for i from 1 to n do
    eqns:=eqns union {int(eval(pde)*phi(i,x,y),[x=-1..1,y=-1..1])};
    vars:=vars union {a[i]}; end do; if bol then print(eval(eqns));
    print(eval(vars)); end if; sol:=solve(eqns,vars);
    if bol then print(sol); end if;
    assign(sol);
  end proc;
```


Определяем полиномиальные базисные функции, удовлетворяющие граничным условиям:

> phi:=(n,x,y)->(1-x^2)^n*(1-y^2)^n:

Определяем аппроксимирующую функцию:

> W:=(x,y,N)->sum(a[i]*phi(i,x,y),'i'=1..N):

Задаем количество членов в аппроксимации и выполняем расчет

> N:=2:

> eq:=value(subs(u(x,y)=W(x,y,N),pde)):

> Bubnov_Gal(eq,N,a,true);

$$\left\{ -\frac{131072}{33075} a_2 - \frac{2048}{525} a_1 + \frac{256}{225}, -\frac{2048}{525} a_2 - \frac{256}{45} a_1 + \frac{16}{9} \right\}$$
$$\{a_1, a_2\}$$

$$\left\{ a_1 = \frac{37}{104}, a_2 = -\frac{105}{1664} \right\}$$

Вычислим скорость на оси канала и безразмерный расход (расход определяется как интеграл от функции скорости по области сечения)

> value(W(x,y,N));evalf(W(0,0,N));

0.29267

> Int(Int(u(x,y),x=-1..1),y=-1..1)=evalf(Int(Int(W(x,y,N),x=-1..1),y=-1..1));

$$\int_{-1}^1 \int_{-1}^1 u(x, y) dx dy = 0.56068$$

Результат совпадает с книгой К.Флетчера «Численные методы на основе метода Галеркина» стр. 26, таб. 1.6. Точные значения приведены в моей книге «Курс математической физики с использованием пакета Maple» (СПб.: Изд-во «Лань», 2016, стр. 457)

$$u(0,0) = 0.2947, \int_{-1}^1 \int_{-1}^1 u(x, y) dx dy = 0.5623$$

С помощью следующих команд изобразим несколько изолиний (рис. 1)

```
> W_b:=value(W(x,y,N)):
myskyblue := ColorTools:-Color([0.8, 0.9, 1.]):
pic3:=implicitplot([W_b=0.01,W_b=0.1,W_b=0.2,W_b=0.25],
x=-1..1,y=-1..1,grid=[50,50],color=khaki,
font=["Helvetica",14],labelfont=["Courier",14]),
polygonplot(ngon,font=["Helvetica",14],color=myskyblue):
pict:=textplot({[0., 1.1, "u = 0.01"],
[0., 0.9, "u = 0.1"],[0., 0.7, "u = 0.2"],
[0., 0.5, "u = 0.25"]}):
display([pic3,pict],axes=frame);
```




Рис. 1. Изолинии скорости.




Рис. 2. Градиент векторного поля.

Изобразим график градиента векторного поля (рис. 2)

```
> plots[gradplot](W_b,x=-1..1,y=-1..1,
  grid=[10,10],style=PATCHNOGRID,arrows=SLIM,
  color=blue,axes=BOXED,symbolsize=14,
  font=[Times,roman,14],labelfont=[Helvetica,roman,14]);
```

4. Конформное отображение. Метод Ритца

Конформные отображения имеют многочисленные приложения в механике, физике и технике, прежде всего для расчетов плоских гармонических векторных полей в гидро- и аэродинамике, теории фильтрации, теории электрических и магнитных полей, теории теплопередачи, теории упругости. Многие стационарные задачи математической физики сводятся к решению простейшего из уравнений эллиптического типа — уравнения Лапласа. Искомая функция, решающая поставленную проблему, определяется областью, в которой она ищется, и граничными условиями, которые должны быть выполнены на контуре этой области. Для простых областей таких, как круг, прямоугольник, круговое кольцо и т. п., задача может быть сравнительно просто решена при относительно сложных граничных условиях. Если же область имеет сложное строение, то решение задачи непосредственно для этой области представляется затруднительным даже для такой простой задачи, как, например, задача Дирихле.

Один из подходов к решению такой задачи — попытаться предварительно преобразовать заданную область к простейшему виду. При таком преобразовании будет меняться, вообще говоря, область, для которой ищется решение, граничные условия, и дифференциальное уравнение, которому должна удовлетворять искомая функция.

Поэтому, большое значение будет, очевидно, иметь такое преобразование, относительно которого само дифференциальное уравнение остается инвариантным. Как известно, уравнение Лапласа остается инвариантным относительно конформных преобразований, осуществляемых аналитическими функциями комплексной переменной. Задача построения функции, отображающей одну область на другую, даже в простом случае, когда одна из областей является кругом, часто вызывает непреодолимые затруднения. Поэтому вызывают интерес методы приближенного построения отображающей функции. Как правило, пытаются аппроксимировать искомую функцию линейной комбинацией хорошо изученных функций. Простейшими и часто используемыми функциями являются многочлены. Известно, что всегда можно построить многочлен, как угодно мало отличающийся от функции (в любой внутренней части области), отображающей область на круг. Кроме того, если граница области простая кривая, то можно построить многочлен, отличающийся от отображающей функции как угодно мало всюду в области, включая и границу. Известно, что среди функций, осуществляющих конформное преобразование заданной односвязной однолистной области D на некоторую другую область, функция, отображающая на круг, обладает экстремальными свойствами: 1) свойством минимума площади при преобразовании области на круг, 2) свойством минимума длины контура при преобразовании области на круг. Указанные свойства можно использовать при построении вариационного метода приближенного конформного преобразования области на круг.

4.1. Свойство минимума площади при преобразовании области на круг.

Рассмотрим функцию

$$\zeta = f(z) = a_0 + a_1 z + a_2 z^2 + \dots$$

Пусть требуется построить конформное отображение однолистной области на круг. При этом потребуем, чтобы заданная внутренняя точка области (всегда можно считать эту точку началом координат) переходила в центр круга. Потребуем также, чтобы определенное направление в этой точке, принимаемое за положительное направление вещественной оси, совпадало бы с положительным направлением вещественной оси преобразованной области. Очевидно, эти условия равносильны требованиям:

$$f(0) = 0, f'(0) = a_1 > 0.$$

Можно считать, что $a_1 = 1$; для этого достаточно рассмотреть функцию $f(z)/a_1$, которая будет так же давать отображение D на круг, но только может быть другого радиуса, что не существенно. Поэтому будем предполагать:

$$f(0) = 0, f'(0) = 1 \tag{8}$$

Функцию, дающую требуемое конформное преобразование, можно искать как решение следующей вариационной задачи: среди функций, аналитических в области D и удовлетворяющих условиям (8), найти ту, которая дает минимум функционалу

$$I = \int_D f'(z) \overline{f'(z)} dx dy \quad (9)$$

по сравнению со всеми другими функциями, которые удовлетворяют требованиям (8). Функционал (9) определяет площадь преобразованной области. Из-за трудностей, возникающих при реализации данного метода, полноценно использовать его стало возможным лишь с появлением систем символьных вычислений таких, как Maple или Mathematica.

4.2 Свойство минимума длины контура при преобразовании области на круг

Рассмотрим всевозможные конформные преобразования области D , совершаемые при помощи аналитических в области D функций $\zeta = f(z)$, удовлетворяющих условиям (8). Вариационную задачу можно сформулировать так: среди функций, аналитических в области D и удовлетворяющих условиям (8), найти ту, которая дает минимум функционалу

$$I = \int_L |f'(z)| ds \quad (10)$$

Функционал (10) выражает длину контура L .

4.3 Применение метода Ритца

Для решения задачи минимизации функционалов (11) и (12) будем использовать метод Ритца. При вычислениях по методу Ритца зачастую получаются плохо обусловленные системы уравнений. Приближенное решение таких систем известными методами приводит к большим погрешностям, что делает результат вычислений бесполезным. Однако применение систем символьных вычислений позволяет эффективно использовать метод, устранить проблемы, связанные с плохой обусловленностью систем, и получить достаточно точное решение задачи. Среди таких систем символьных вычислений стоит отметить две наиболее мощные системы — это Maple и Mathematica.

Заметим, что используя свойство минимума площади при отображении области на круг, можно искать не непосредственно функцию $f(z)$, а ее производную $f'(z)$, так как под знак интеграла (9) функция $f(z)$ не входит, и по найденной производной $f'(z)$ восстановить функцию $f(z)$ по формуле

$$f(z) = \int_0^z f'(z)dz.$$

Используя свойство минимума длины контура при отображении области на круг, при вычислениях удобнее искать приближения не непосредственно к производной $f'(z)$, в виду линейности подынтегрального выражения относительно $|f'(z)|$, а к величине $\sqrt{f'(z)}$. Допустимыми будут все функции $\varphi(z)$, обращающиеся в единицу в начале координат. Среди них будем искать ту, которая дает наименьшее значение функционалу

$$I = \int_L |\varphi(z)|^2 ds = \int_L \varphi(z) \overline{\varphi(z)} ds \quad (11)$$

Если удастся найти решение вариационной задачи для функционала (11), то функция совершающая искомое отображение, определится по формуле

$$I = \int_0^z \varphi^2(z) dz.$$

5. Примеры построения конформных отображений

Рассмотрим применение метода Ритца на конкретных примерах в системе Maple.

- * **Задача 1.** Приблизительно построить конформное отображение эллипса с полуосями $a = 3$, $b = 2$; начало координат — в точке пересечения осей эллипса (рис. 3).

Все вычисления выполняем в Maple. Подключаем необходимые пакеты.

```
> with(VectorCalculus): with(Optimization):
```

Для наглядности можно изобразить отображаемую область (рис. 3); в данном случае это — эллипс. Набираем команду:

```
> plot([3*cos(t),2*sin(t),t=0..2*Pi],color=blue,  
       scaling=constrained,gridlines=true);
```

После нажатия клавиши «Enter» на экране монитора появится изображение эллипса (рис. 3).




Рис. 3. Эллипс.

Задаем степень полинома n и выражения для $f'(z)$ и $f''(z)$ — переменные P и P_1 , вычисляем подынтегральную функцию в функционале (9) — переменная Q .

> $n := 4$:

$P := 1 + \sum_{i=1}^n ((a[i] + I * b[i]) * (x + I * y)^i, i = 1 .. n):$

$P_1 := 1 + \sum_{i=1}^n ((a[i] - I * b[i]) * (x - I * y)^i, i = 1 .. n):$

$Q := \text{expand}(P * P_1):$

Вычисляем функционал (11)

> $J := \text{int}(Q, [x, y] = \text{Ellipse}((1/9) * x^2 + (1/4) * y^2 - 1)):$

Минимизируем функционал:

> $\text{res} := \text{Minimize}(J):$

Результат выполнения последней команды появляется на экране монитора:

```
res:= [16.787, [a1 = 0., a2 = -0.985e - 1, a3 = 0.,  
a4 = 0.438e - 2, b1 = 0., b1 = 0., b1 = 0., b1 = 0.]]
```

Далее надо присвоить вычисленные значения коэффициентам a_i и b_i ; это осуществляется командами

```
> op(2, res): assign(%):
```

Таким образом, найдена производная отображающей функции $f'(z)$ — обозначим ее p_1 :

```
> p1:= 1+sum((a[i]+I*b[i])*z^i, i = 1 .. n);  
p1:= 1 + (-0.0985 + 0.I)z2 + (-0.00438 + 0.I)z4
```

Восстанавливаем отображающую функцию

```
> p:= int(p1, z = 0 .. z);  
p:= z - 0.0328z3 + 0.000875z5
```

Для последующего использования удобно преобразовать полученное выражение в функцию


```
> p:= unapply(p, z):
```

и функцию — отображение на единичный круг

```
> w:= unapply(p(z)/abs(p(3)), z):
```

Для проверки качества полученного решения параметризуем границу области D и построим график ее образа при отображении функцией $w(z)$

```
> z1 := 3*cos(t)+(2*I)*sin(t):  
pic1 := plot([evalc(Re(w(z1))), evalc(Im(w(z1))), t = 0 .. 2*Pi]):  
pic2 := plot([cos(t), sin(t), t = 0 .. 2*Pi], color = blue):  
plots[display]({pic1, pic2})
```



На рисунке голубым цветом показан точный единичный круг, красным цветом — приближение к нему.

Как видим, уже многочлен пятой степени дает достаточно точное приближение к кругу. Можно оценить и погрешность вычислений

> assume(t, real):

absw1 := abs(expand(w(z1))):

Maximize(abs(absw1-1), {0 <= t, t <= 2*Pi});

$[0.18152e - 2, [t = 1.02918]]$

Погрешность в данном случае составляет не более, чем $\varepsilon = 0.002$.

Следует отметить, что точность и трудоемкость решения задачи зависит от характера отображаемой области. Так для отображения на единичный круг эллипса с большим эксцентриситетом, например, с полуосями $a = 4$, $b = 1$ и эксцентриситетом $\eta = \frac{\sqrt{15}}{4} \approx 0.968$ приходится использовать комплексный многочлен более высокой степени. Результат решения задачи при степени многочлена $n = 15$ дает погрешность не более чем $\varepsilon = 0.02$, т.е. на порядок хуже, чем в предыдущем случае.

В заключение приведем процедуру построения комплексного полинома, осуществляющего конформное отображение треугольника на круг и использующую свойство минимума длины контура:

```

> ConformPolTriangl:=proc(n,A1,B1,C1) local P,P1,Q,J1,J2,J3,J,res,A,B;
global p, w;
P:=1+sum((A[i]+I*B[i])*(x+I*y)^i,i=1..n);
P1:=1+sum((A[i]-I*B[i])*(x-I*y)^i,i=1..n);
Q:=expand(P*P1);
J1:=PathInt(Q,[x, y]=Line(A1,B1));
J2:=PathInt(Q,[x,y]=Line(B1,C1));
J3:=PathInt(Q,[x, y]=Line(C1,A1));
J:=J1+J2+J3;res:=Minimize(J);op(2,res);assign(%);
P:=subs(x+I*y=z,P);
p:=simplify(int(P^2,z=0..z));
p:=unapply(p,z);w:=unapply(p(z)/abs(p(A1[1]+I*A1[2])),z)
end proc:

```

С помощью этой процедуры построим отображение вытянутого треугольника на круг.

Задаем вершины треугольника:

```
> x1 := -1; x2 := 1; x3 := 0; y1 := -2; y2 := 3; y3 := -2;
```

```
A1 := '<,>'(x1, y1); B1 := '<,>'(x2, y2); C1 := '<,>'(x3, y3);
```



Изобразим этот треугольник с помощью команды:

```
> display(polygon([[ -1,-2],[1,3],[0,-2]]),color=green,  
  thickness=2,scaling=constrained);
```

Решаем задачу:

```
> Digits:=100: ConformPolTriangl(60,A1,B1,C1):
```

Ниже на рисунках показан результат решения задачи



Рассмотренные примеры приближенного построения конформных отображений областей на круг единичного радиуса показывают эффективность применения вариационного метода в совокупности с системой аналитических вычислений Maple к решению указанных задач.